Visual Basic i API programiranje

Uvod

Sigurno ste se mnogo puta upitali kako da iz Visual Basic-a dospete do srži operativnog sistema, uradite više nego što Visual Basic u suštini dopušta ili da naprimer kontrolišete neki sopstveni hardver preko računara a da pritom ne napuštate udobno VB razvojno okruženje? Iako možda zvuči kao nemoguće postoji način da se to uradi. Odgovor je – upotrebite API. 
Šta je API?

Kao VB programer Vi ste već koristili API (barem indirektno) jer svaki Windows program koristi API pozive koje sistem koristi da bi uradio ono što očekujete. API (Application Programming Interface) je set unapred definisanih Windows funkcija koje se koriste za kontrolu izgleda i ponašanja svakog Windows elementa. Te funkcije enkapsuliraju funkcionalnost Windows okruženja tako da se u izvesnom smislu može reći da je API zapravo native kôd za Windows. Visual Basic se ponaša kao shell za API jer se svaka linija kôda konvertuje u ekvivalentan API poziv. API, u izvesnom smislu, podseća na često korišćeno referenciranje u Visual Basic-u sa tom razlikom što se na biblioteku ne referencirate direktno već posebno deklarišete funkcije ili rutine koje biblioteka poseduje u sebi i koristite ih kao da su deo Visual Basic-a. 

Gde je API?

Glavnina API funkcija je enkapsulirana u setu .dll fajlova - kernel32.dll, user32.dll, gdi32.dll, shell32.dll itd. Pomenute dinamičke biblioteke se nalaze na svim Windows mašinama (barem na onim koje mogu da startuju () i dokle god koristite ove .dll-ove ne morate ništa dodatno da instalirate uz Vašu aplikaciju što svakako ima uticaja na veličinu distribucionog paketa i eventualno na brzinu aplikacije. Sledeća tabela daje neke od opisa biblioteka.

Tabela 1.0 Opis nekih dinamičkih bibliteka sadržanih u Windows operativnom sistemu

	Advapi32.dll
	Napredni API servisi podrška security i Registry bazi

	Comdlg32.dll
	Common dialog API 

	Gdi32.dll
	Grafički interfejs API 

	Kernel32.dll
	Core Windows 32-bit base API podrška

	Lz32.dll
	32-bit kompresione rutine

	Mpr.dll
	Multiple Provider Router library

	Netapi32.dll
	32-bit Network API library

	Shell32.dll
	32-bit Shell API library

	User32.dll
	User interface rutine

	Version.dll
	Version library

	Winmm.dll
	Windows multimedia library

	Winspool.drv
	Print spooler interface sadrži print spooler API pozive



Kako se koriste API funkcije iz Visual Basic-a?

Kao što je pomenuto ranije te funkcije su enkapsulirane u spoljne biblioteke tako da se ne mogu koristiti iz Visual Basic-a direktno. Pre korišćenja je neophodno da ih deklarišemo. Deklaracija se može obaviti u generalnoj sekciji modula ali ako je forma u pitanju onda ta deklaracija mora biti Private jer Visual Basic ne dopušta Public Declare izraz u formi. Stoga je uobičajeno da se API deklariše u standardnom modulu (.BAS) i da se odatle poziva kroz ceo projekat što eliminiše dupliranje kôda i omogućava mnogo lakše održavanje. Najčešće se dodatna API funkcionalnost realizuje kroz klase zbog ponovnog korišćenja u nekim kasnjim projektima.

Zašto se koristi API u Visual Basic-u?

Zato što već postoji na računaru. Kada uporedite sve ono što operativni sistem radi sa onim što može da se uradi iz Visual Basic-a razlika je očigledna. Visual Basic je posebno ograničen u sistemskom programiranju tako da korišćenje API-ja pomaže u prevazilaženju tih ograničenja što je mnogo brže rešenje nego da počnete da učite npr. VisualC++. Čak i da se odlučite za učenje C/C++ opet ćete koristiti API jer je neproduktivno pisati ono  što već postoji. Drugi razlog je to što su programi pisani u Visual Basic-u sporiji od ekvivalenata pisanih u C++ ili Delphi-ju. To je cena lakšeg debug-ovanja, lakšeg i bržeg pisanja kôda i bržeg razvojnog ciklusa. Kako je API interfejs Windows-a, iz toga proizilazi i da je brži od klasičnog VB kôda (videćemo kasnije u kojim slučajevima i zašto je to tako) a kako su dinamičke biblioteke prisutne na svim Windows mašinama proizilazi i da je distribucioni paket manji. Sigurno je da će Vam API zatrebati da odredite na kom disku je instaliran operativni sistem kako biste odredili neku putanju za Vašu aplikaciju, da iz Visual Basic-a sakrijete Vaš program u task listi i slično. 

Kada je brzina u pitanju treba raspraviti neke detalje vezane za ovu temu. API je, po iskustvu mnogih programera, brži od Visual Basic-a ali ta brzina generalno ne potiče od API-ja kao nekakvog superiornijeg rešenja već je ta brzina u većini slučajeva zapravo posledica neoptimizovanog kôda u Visual Basic-u. Drugim rečima, Visual Basic je u redu kada je brzina u pitanju samo je uobičajeno da se malo vremena u razvoju aplikacije daje finoj optimizaciji kôda. Jedini izuzetak od ovog pravila je grafika. Na tom polju je API brži od Visual Basic-a. 
 
Nedostatci API-ja u odnosu na Visual Basic
Dosta je teži za korišćenje od Visual Basic-a. To, između ostalog, znači i da je ekstremno netolerantan na greške programera. Veoma je lako srušiti sopstvenu aplikaciju, blokirati neke ili sve ostale aplikacije koje su pokrenute ili čak blokirati ceo sistem sa pogrešnim API pozivom. Da stvar bude još komplikovanija, pogrešan API poziv može da toliko utiče na sistem da čak i nakon restartovanja, mašina neće da se pokrene (ova varijanta se dešava ako pokušavate preko API-ja da pišete ili pogotovu brišete ili menjate Registry unose a da pri tome niste sigurni kakav to uticaj može imati na sistem - zato je veoma pametno da pre eksperimentisanja sa ovom vrstom API-ja eksportujete sadržaj Registry baze da biste kasnije sistem mogli da vratite u pređašnje stanje). Drugi scenario za spektakularni pad sistema ili aplikacije bi bio kada neispravan API poziv poziva još desetine drugih pa su onda rezultati potpuno nepredvidivi. Ipak, ovi nedostatci nikako ne smeju da vas spreče u korišćenju API-ja. Jedino što je potrebno je da vodite računa o mogućim posledicama. U krajnjoj liniji, reinstalacija Windows-a u najgorem slučaju i bekap Registy baze i podataka su tri osnovne stvari koje bi svaki programer morao da zna i primenjuje a one se veoma dobro uvežbavaju sa (pogrešnim) API-jem. Budite opušteni – plavi ekrani se dešavaju i onima koji se ne bave programiranjem. Za one koji žele da igraju na sigurno web je odlična referentna baza podataka koja će definitivno pomoći u razumevanju API- funkcija a uvek imate i F1 taster na tastaturi (MSDN, Widnows SDK - Software Development Kit) kao i raznu third party (nezavisni proivođači – npr. Desaware Inc.) dokumentaciju. Potražite preko nekog od web pretraživača API i dobićete nMb dokumenata, objašnjenja i pomoći.


Kako radi API?
Videćete, recimo u početku… komplikovano. Visual Basic koristi API da bi pozvao API funkciju. Kada pozovete neku API funkciju Visual Basic pomoću API funkcije LoadLibrary prvo snimi strukturu modula koji sadrži deklarisane funkcije. Ako nešto nije u redu sa pozivom Visual Basic će se pobuniti uz poruku o grešci. Zatim se poziva funkcija GetProcAddress koja vraća početnu adresu u memoriji koju .dll treba tek da dobije u memoriji. Nakon toga, Visual Basic poziva Windows sistem za praćenje grešaka koristeći API funkciju SetErrorMode. Posle ove pripremne faze, Visual Basic postavlja nit i poziva deklarisanu funkciju. Zatim se izvršava pozvana funkcija i nakon izvršenja funkcije se oslobađa nit. U slučaju da je nešto krenulo naopako, popunjava se svojstvo LastDLLError objekta Error kôdovima grešaka. Potom se kopiraju vrednosti argumenata u promenljive. Konačno, iz memorije se uklanja struktura modula, memorija se oslobađa (garbagge collection) i kontrola se vraća Visual Basic-u. Iz strategije pristupa se vidi da je postupak dug ali je fleksibilan jer se svi API pozivi kasno vezuju (late binding). Naravno, još uvek ostaje nedostatak u dužini postupka ali sa druge strane Visual Basic u svojim izvršnim bibliotekama sadrži funkcije koje se izvršavaju direktno bez pozivanja drugih funkcija pa bi to bio razlog da aplikaciju optimizujete direktno iz Visual Basic-a a ne preko API-ja. Jedini izuzetak za API optimizaciju bi bile grafičke operacije koje su, kao što je opšte poznato, same po sebi intenzivne u pogledu procesorskih i memorijskih resursa. 
Windows API: prozori, događaji i poruke
U operativnoj logici Windows operativnih sistema sve je prozor. Forma, kontrola, ikona, padajući meni, komandno dugme... sve su to zapravo prozori. Da bi upravljao tim elementima, operativni sistem svakom od njih dodeljuje jedinstaveni identifikator u obliku broja koji se zove window handle (ručica prozora) ili hWnd. OS stalno prati aktivnost u tim prozorima. Kada se desi neki događaj (naprimer korisnik je kliknuo na komandno dugme) šalje se poruka operativnom sistemu koji reaguje na tu poruku. 


Šta sve može da se uradi sa API-jem?

Ovo je veoma široka tema. U principu odgovor glasi : sve ono što ne može direktno kroz Visual Basic a da pri tome ostanemo na razumno visokom nivou programiranja. Mogu se kontrolisati miš, tastatura, monitor, memorija, operativni sistem, korisnici, Registry baza i još mnogo toga. U principu API funkcijom ili funkcijama možete uraditi sve ono što se na računaru može uraditi. Jedino Vam treba prava funkcija i ispravna deklaracija i korišćenje. Za manipulaciju Windows-a (sa kojom ćemo početi) pomoću API funkcija glavna alatka je API Text Viewer koji se nalazi u Microsoft Visual Studio 6.0 Tools –ima a može se pokrenuti i sa komandne linije uz upotrebu odgovarajućih parametara a možete ga koristiti iz VB okruženja kao Add-In.

Slika1.0 - API Text Viewer

Ako do sada niste odustali onda je vreme da podhitno uradimo nešto kako bih uspeo da zadržim Vašu pažnju. Pokrenite API text Viewer naprimer kao samostalnu aplikaciju (Programs/Microsoft Visual Studio/Microsoft Visual Studio6.0 Tools/API Text Viewer). Sa menija File odaberite opciju Load Text File i kada se pojavi standardni Open Dijalog odaberite WIN32API.TXT (obavezno ga pogledajte i iz NotePad-a kada se izveštite u API programiranju pa će Vam mnoge na prvi pogled nejasne stvari biti jasnije). Iz liste Available Items selektujte stavku GetComputerName. Pritisnite komandno dugme Add i u tekst box-u će se pojaviti tekst sa slike 1.0. Komandnim dugmetom Copy kopirajte sadržaj u Clipboard. Sada u Clipboard-u imate Deklaraciju API funkcije GetComputerName koja će nam vratiti ime računara. Ovde treba pomenuti da WIN32API.TXT sadrži greške u deklaracijama što može biti veoma frustrirajuće pa to treba imati u vidu kada radite sa njim a rezultati uporno izostaju. Na Internetu i u MSDN se mogu naći ispravke nekih od tih grešaka.

Primer 1:
Pokrenite VB i kreirajte nov standard EXE projekat. Pored forme koja već postoji u projektu dodajte standardni modul i u General /Declarations sekciji modula dodajte (Ctrl+V) sledeći kôd:


Public Declare Function GetComputerName Lib "kernel32" Alias_

 "GetComputerNameA" (ByVal lpBuffer As String, nSize As Long) As Long

zatim tu deklaraciju preuredite da izgleda kao donja (samo zamenite GetComputerName sa ImeRacunara - objasniću kasnije zašto)


Public Declare Function ImeRacunara Lib "kernel32" Alias_

 "GetComputerNameA" (ByVal sBuffer As String, lSize As Long) As Long

a u formu dodajte sledeći kôd:

Private Sub Form_Load()

Dim strImeKompjutera As String

strImeKompjutera = String(255, Chr(0))

ImeRacunara strImeKompjutera, 255

strImeKompjutera = Left(strImeKompjutera, InStr(1, strImeKompjutera,_

 Chr(0)) - 1)

MsgBox strImeKompjutera

End Sub

Pokrenite program i u MsgBox-u bi morali da dobijete ime Vašeg računara (koje možete proveriti sa Desktop-a / desni klik na Network Neighborhood ili My Network Places pa zatim na kartici Identification). Ovo je zgodan trenutak da malo prodiskutujemo o kôdu. Kao prvo deklaracija…

Public Declare Function ImeRacunara Lib "kernel32" Alias "GetComputerNameA" (ByVal sBuffer As String, lSize As Long) As Long

… Public je javna deklaracija. Da smo deklaraciju smestili u General/Declarations sekciju forme morala bi da počinje sa Private a onda ne bi bila vidljiva iz ostalog dela projekta pa bismo svuda gde bismo trebali ovu funkciju morali ponovo da je deklarišemo što bi kôd učinilo glomaznijim. ImeRacunara je alias funkcije GetComputerNameA. Ovo je zgodna tehnika da neku funkciju pozivamo po imenu koje smo joj mi dodelili i na taj način se mogu izbeći konflikti kada se funkcija koju koristimo zove isto kao neka druga funkcija u projektu  ili je (češći razlog) naziv funkcije u koliziji sa konvencijom programskog jezika koji koristimo (npr. počinje karakterom _  ). Kernel32 je ime dinamičke biblioteke koja, između ostalih, sadrži funkciju GetComputerNameA i načelno bi trebalo da glasi Kernel32.dll ali se ekstenzija može izostaviti jer se podrazumeva da je u pitanju .dll. Takođe se može izostaviti putanja do .dll-a jer se podrazumeva da je u System direktorijum Windows-a. Ove napomene su bitne kada koristite Windows API. Takođe možete koristiti i API-je od drugih proizvođača ali u tom slučaju morate navesti stvari koje smo izostavili u ovom primeru.
Argumenti

Ova funkcija koristi dva argumenta. Prvi je string i prosleđuje se po vrednosti (ByVal) a ne po referenci (ByRef) - morate znati razliku između ova dva načina prosleđivanja argumenata. Na brzinu ću Vas podsetiti: ByVal prosleđuje tačnu vrednost i ne menja vrednost argumenta dok ByRef  (podrazumevana vrednost u Visual Basic-u) prosleđuje zapravo referencu na promenljivu. 

LpBuffer - ukazuje na buffer koji sadrži ime računara. Promenljiva je tipa string.

nSize – pokazuje na promenljivu koja određuje maksimalnu dužinu buffer-a. Ta vrednost mora da bude dovoljno velika da sadrži MAX_COMPUTERNAME_LENGTH + 1 karakter. Kada je ime računara u pitanju radi s o 1 bajtu pa je stoga decimalni ekvivalent 255. 
Ostatak kôda …

Microsoft upozorava da ako ne popunimo buffer to može dovesti do pada aplikacije. Meni se to nije desilo – verovatno neće ni Vama. Možete čak i probati jer ovo i nije neki, po posledicama, ozbiljan API tako da osim eventualnog pada aplikacije ne bi smelo da se desi ništa loše. U primeru je to urađeno tako što je buffer popunjen sa Chr(0) a to je po ASCII tabeli null karakter. Posle toga smo u promenljivu strImeKompjutera upisali ime kompjutera a zatim smo poskidali višak karaktera Chr(0) i na kraju smo kroz MsgBox funkciju prikazali sadržaj promenljive strImeKompjutera. U principu veoma prosto – iskoristili smo funkciju biblioteke Kernel32.dll koja nam je dala ime računara. Sada pogledajte sledeću činjenicu: u WIN32API-ju ima oko 1500 funkcija plus MAPI (Messaging API) plus TAPI (Telephony API) plus razni API drugih proizvođača – veoma brzo dolazimo do cifre od nekoliko hiljada API funkcija kojima se može uraditi mnogo toga što direktno iz Visual Basica ne može ili je mnogo lakše koristeći API. Sledeći primer će baš ilustrovati ovu drugu varijantu - lakše uz pomoć API-ja zbog enkapsulacuje.

Primer 2:

Iz Visual Basic-a možete napraviti aplikaciju koja će okrenuti neki telefonski broj preko modema. Za sada cela stvar ne deluje naročito kompikovano međutim tek kada se upustite u posao shvatićete koliko stvar može da bude komplikovana. U MSDN-u postoji biblioteka primera i evo primera u kome nemojte zalaziti mnogo u detalje – u MSDN-u je priložen čisto kao ilustarcija kako se koristi MSComm kontrola (a mi ćemo ovaj primer iskoristiti da bi uočili  potencijalne probleme) – zato blago pregledajte kôd i pređite na kraj kôda radi dalje diskusije :

Option Explicit

'preuzeto iz MSDN-a, ostatak komentara je uklonjen zbog prostora

DefInt A-Z

' This flag is set when the user chooses Cancel.

Dim CancelFlag

Private Sub CancelButton_Click()

    ' CancelFlag tells the Dial procedure to exit.

    CancelFlag = True

    CancelButton.Enabled = False

End Sub

Private Sub Dial(Number$)

    Dim DialString$, FromModem$, dummy

    ' AT is the Hayes compatible ATTENTION command and is required to send 'commands to the modem.

    ' DT means "Dial Tone." The Dial command uses touch tones, as opposed 'to pulse (DP = Dial Pulse).

    ' Numbers$ is the phone number being dialed.

    ' A semicolon tells the modem to return to command mode after dialing  '(important).

    ' A carriage return, vbCr, is required when sending commands to the modem.

    DialString$ = "ATDT" + Number$ + ";" + vbCr

    ' Communications port settings.

    ' Assuming that a mouse is attached to COM1, CommPort is set to 2

    MSComm1.CommPort = 2

    MSComm1.Settings = "9600,N,8,1"

        ' Open the communications port.

    On Error Resume Next

    MSComm1.PortOpen = True

    If Err Then

       MsgBox "COM2: not available. Change the CommPort property _

to another port."

       Exit Sub

    End If

' Flush the input buffer.

    MSComm1.InBufferCount = 0

' Dial the number.

    MSComm1.Output = DialString$

 ' Wait for "OK" to come back from the modem.

    Do

       dummy = DoEvents()

' If there is data in the buffer, then read it.

       If MSComm1.InBufferCount Then

          FromModem$ = FromModem$ + MSComm1.Input

          ' Check for "OK".

          If InStr(FromModem$, "OK") Then

             ' Notify the user to pick up the phone.

             Beep

             MsgBox "Please pick up the phone and either press Enter or click OK"

             Exit Do

          End If

       End If

' Did the user choose Cancel?

       If CancelFlag Then

          CancelFlag = False

          Exit Do

       End If

    Loop

 ' Disconnect the modem.

    MSComm1.Output = "ATH" + vbCr

  ' Close the port.

    MSComm1.PortOpen = False

End Sub

Private Sub DialButton_Click()

    Dim Number$, Temp$

    DialButton.Enabled = False

    QuitButton.Enabled = False

    CancelButton.Enabled = True

 ' Get the number to dial.

    Number$ = InputBox$("Enter phone number:", Number$)

        If Number$ = "" Then Exit Sub

    Temp$ = Status

    Status = "Dialing - " + Number$

 ' Dial the selected phone number.

    Dial Number$

DialButton.Enabled = True

    QuitButton.Enabled = True

    CancelButton.Enabled = False

 Status = Temp$

End Sub

Private Sub Form_Load()

    ' Setting InputLen to 0 tells MSComm to read the entire

    ' contents of the input buffer when the Input property

    ' is used.

    MSComm1.InputLen = 0

End Sub

Private Sub QuitButton_Click()

    End

End Sub
 

Kao što ste mogli videti iz gornjeg kôda potrebno je da poznajete sledeće stvari da bi ste napisali prethodni program: ATX komande modema (u smislu da li ih podržava ili ne - kako su ovo osnovne komande nije problem ali probajte da zamislite malo komplikovaniju stvar), zatim port na kome se nalazi modem što morate unapred znati ili ga morate pronaći u Registry bazi što može biti komplikovano. Čitava stvar već počinje da se komplikuje i pitanje je gde ćete kao programer naići na probleme. Dalje morate postaviti parametre MSComm kontrole i na kraju kada se izborite sa svim tim detaljima moraćete da u distribucioni paket uključite i samu ActiveX kontrolu koja, priznajem, i nije nešto posebno velika da bi dramatično povećala veličinu instalacionog paketa. Za kraj još jedan, možda najbitniji, detalj - ova aplikacija će vrlo verovatno sporije raditi od sledećeg primera zbog količine kôda koji treba da se prevede i izvrši. Međutim, evo API ekvivalenta koji će odraditi isti posao brže i za Vas lakše.

Option Explicit

‘deklaracija i konstante

Private Declare Function Pozovi& Lib "TAPI32.DLL" Alias_ 

"tapiRequestMakeCall" (ByVal DestAddress$, ByVal AppName$, ByVal_

 CalledParty$, ByVal Comment$)

Private Const Ne_mogu_da_pokrenem_Windialer = -2&

Private Const Pun_buffer_zahteva = -3&

Private Const Neispravan_broj_telefona = -4&

‘pozivanje broja upisanog u tekst box txtBroj

Private Sub cmdPoziv_Click()

Dim poruka As String

Dim rezultat As Long

rezultat = Pozovi&(Trim(txtBroj), "", "", "")

‘ako se pojavila greska...

If rezultat <> 0 Then

        poruka = "Greska u aplikaciji : "

Select Case rezultat

Case Ne_mogu_da_pokrenem_Windialer

poruka = poruka & "Ne mogu da pokrenem Dialer"

Case Pun_buffer_zahteva

poruka = poruka & "Trenutno zauzeta aplikacija"

Case Neispravan_broj_telefona

poruka = poruka & "Nevazeci broj telefona"

Case Else

poruka = poruka & "Nepoznata greska"

End Select

        MsgBox poruka

End If

End Sub

‘unosenje broja od strane korisnika

Private Sub txtBroj_Change()

‘ako je duzina teksta > 0 onda je omoguceno pozivanje broja

cmdPoziv.Enabled = Len(Trim(txtBroj)) > 0

End Sub

Kao što se može videti iz priloženog treba samo deklarisati funkciju 

Private Declare Function Pozovi& Lib "TAPI32.DLL" Alias_

 "tapiRequestMakeCall" (ByVal DestAddress$, ByVal AppName$, ByVal_

 CalledParty$, ByVal Comment$)

i treba razmotriti parametre koji se prosleđuju. Ostatak kôda je obična zaštita od pogrešnog korišćenja aplikacije od strane korisnika. Korisniku je onemogućeno da pozove broj a da ga pre toga nije uneo. Funkcija tapiRequestMakeCall vraća vrednost različitu od nule ako je došlo do neke greške. Imena konstanti u primeru (inače označavaju broj greške) su takva da je evidentno kakvu grešku označavaju te je nepotrebno posebno ih objašnjavati. Jedino što bi eventualno trebalo objasniti su argumenti. Kao prvo svi su prosleđeni po vrednosti. DestAddress$ (logički mada ne i funkcionalno obavezan parametar) je broj koji treba pozvati dok su ostali parametri opcioni i odnose se na log fajl modema. AppName$ označava ime aplikacije iz koje je pokrenuta funkcija, CalledParty$ je ime vlasnika tel. broja i Comment$ je nekakav komentar o telefoniranju. U zavisnosti od operativnog sistema moguće su varijacije u ponašanju API funkcija. Na primer, prethodna ilustracija pod W9X-Me-W2K ne upisuje sve opcione parametre u log fajl (upisuje samo CalledParty$) dok pod W2K u Dialer aplikaciji  parametre AppName$ i Comment$ upisuje direktno u formu Dialer aplikacije.

Takođe nikada ne treba polaziti od pretpostavke da će API funkcija raditi isto ili slično na različitim operativnim sistemima. Ono što radi pod W9X ne mora raditi pod NT-om i obrnuto. Obavezno testirajte aplikaciju pod svakim operativnim sistemom posebno. 

Za kraj, ovo bi bila mala ilustracija snage API funkcija i njihove upotrebe međutim ona nikako nije pokrila celu oblast API programiranja kao ni koncepte koji se koriste. Ideja je bila da kroz primere steknete opštu predstavu kako API funkcioniše i da se eventualno podstaknete na dalja istraživanja i ideje. Iz Visual Basic-a koristeći MAPI (Messaging API), SAPI (Speech API) i TAPI (Telephony API) se relativno lako (zbog velikog stepena enkapsulacije) može kreirati, recimo, aplikacija koja će pokupiti elektronsku poštu iz mailbox-a na internetu, pozvati Vas telefonom kada se pojavi nova poruka i pročitati Vam je preko telefonske linije, što očigledno prevazilazi mogućnosti samog Visual Basic-a. Međutim, za ovako nešto je ipak potrebno upoznati i WOSA (Windows Open Services Archtecture) model. Stoga, sledeći put pri planiranju aplikacije razmislite o API-ju kao o potencijalnom rešenju.

Dodatak

Primer 3: zatvaranje MsgBox prozora

Kod za formu

Option Explicit

Private Sub Command1_Click()

SetTimer hWnd, ZatvoriMsgBox, 2000&, AddressOf TajmerProcedura 

Call MessageBox(hWnd, "Za dve sekunde se zatvaram!", "Samozatvarajuci_

 MsgBox", ZnakPitanja Or Modalno)

End Sub

Kod za modul

Option Explicit

'konstante

Public Const Zatvori As Long = &H10&

Public Const ZatvoriMsgBox As Long = &H5000&

' MessageBox flagovi

Public Const ZnakPitanja As Long = &H20&

Public Const Modalno As Long = &H2000&

'deklaracije API funkcija

Public Declare Function NadjiProzor& Lib "user32" Alias "FindWindowA"_

 (ByVal lpClassName$, ByVal lpWindowName$)

Public Declare Function PosaljiPoruku Lib "user32" Alias "PostMessageA"_

 (ByVal hWnd&, ByVal wMsg&, ByVal wParam&, lParam As Any) As Long

Public Declare Function MessageBox& Lib "user32" Alias "MessageBoxA"_

 (ByVal hWnd&, ByVal lpText$, ByVal lpCaption$, ByVal wType&)

Public Declare Function SetTimer& Lib "user32" (ByVal hWnd&, ByVal_ 

nIDEvent&, ByVal uElapse&, ByVal lpTimerFunc&)

Public Declare Function KillTimer& Lib "user32" (ByVal hWnd&, ByVal_

 nIDEvent&)

Sub TajmerProcedura(ByVal hWnd&, ByVal uMsg&, ByVal idEvent&, ByVal_

 dwTime&)

'prvo uklanjamo timer da ne bi radio dalje

KillTimer hWnd, idEvent

' vrsta dogadjaja

If idEvent = ZatvoriMsgBox Then 'zatvaramo msgbox      

Dim hMessageBox&

' trazimo messagebox prozor

hMessageBox = NadjiProzor("#32770", "Samozatvarajuci MsgBox")

' i saljemo mu Zatvori poruku

If hMessageBox Then Call PosaljiPoruku(hMessageBox, Zatvori, ByVal 0&,_

 ByVal 0&)

End If

End Sub

 Uradite Copy & Paste kod iz prethodnog primera u nov standard EXE i probajte primer. U projektu trebate jednu formu sa komandnim dugmetom (Command1) i jedan standardni modul. 

Primer 4:  Slike u padajućem meniju forme

Option Explicit

Private Declare Function GetMenu Lib "user32" _

   (ByVal hwnd As Long) As Long

Private Declare Function GetSubMenu Lib "user32" _

   (ByVal hMenu As Long, ByVal nPos As Long) As Long

Private Declare Function SetMenuItemBitmaps Lib "user32" _

   (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long, _

    ByVal hBitmapUnchecked As Long, ByVal hBitmapChecked As Long) As Long

Const pozicija = &H550&

Private Sub Form_Load()

    Dim MeniHandle As Long, promenljiva As Long, SubMeniHandle As Long
    MeniHandle = GetMenu(hwnd)

    SubMeniHandle = GetSubMenu(MeniHandle, 0)

    promenljiva = SetMenuItemBitmaps(SubMeniHandle, 0, pozicija,_

 imOtvori.Picture, imOtvori.Picture)

    promenljiva = SetMenuItemBitmaps(SubMeniHandle, 1, pozicija,_

 imSacuvaj.Picture, imSacuvaj.Picture)

    promenljiva = SetMenuItemBitmaps(SubMeniHandle, 3, pozicija,_

 imStampanje.Picture, imStampanje.Picture)

    promenljiva = SetMenuItemBitmaps(SubMeniHandle, 4, pozicija,_

 imPodesavanje.Picture, imPodesavanje.Picture)

End Sub

Izgled menija

Na formi se nalaze 4 image kontrole sa postavljenim svojstvima Picture (.bmp fajlovi). Slike treba da budu približne veličine kao u meniju. Kontrole su u kodu označene sa prefiksom im.

